

**you-try-it-10.xlsx**  
**Step-by-Step Guide**  
**ver. 02/14/2023**

**Abstract**

This document provides step-by-step instructions for the Excel workbook you-try-it-10.xlsx (saved from Excel 2010). The worksheets contain data for practice exercises keyed to Chapter 10 of:

Brian M. Tissue, *Basics of Analytical Chemistry and Chemical Equilibria*, 2<sup>nd</sup> Ed.  
(John Wiley: New York, 2023).

**Worksheet in the Workbook**

**Page in this Guide**

10.A AAS.....	2
10.B AES .....	9
10.C ICP-MS.....	16

**General Advice**

These exercises are rather different from the equilibrium calculations. Instrumental methods involve performing validation experiments to develop procedures that provide accurate measurements. The analyst must keep an eye open for clues that instrumental results are deviating from the expected behavior. Such clues could be non-linearity (when linearity is expected), outliers, non-repeatable measurements, drift, and deviations in standard addition experiments. These exercises illustrate the type of data obtained in atomic spectrometry and provide practice in plotting and visualizing the data.

**Version History**

11/20/2018	First posting.
02/14/2023	Minor updates to match 2 <sup>nd</sup> Ed. (Note: Ignore Table headings that still refer to Chapter 9.)

For updates visit the text support website:

**<https://www.achem.org/>**


## 10.A AAS

This worksheet contains a table with absorbance measurements for a set of calibration standards and for two unknown samples. We expect a linear response for most instrumental measurements that we make. Flame atomic absorption spectrometry is an exception where nonlinear calibration curves are common. This exercise evaluates the measurement results using two different calibration functions.

Concepts	Concept Synopsis
calibration function	The calibration function should model the actual data. Deviations between the model and data should be random and not systematic.

### 1. Plot the standard data.


Use the mouse to select the data in Table 10.A.1, cells C17:D22. Insert a scatter plot from the Insert ribbon. Add axis labels and adjust the formatting to remove empty space from the plot. Right-click on a data point in the chart and select “Add Trendline...” Choose the Linear function under Trendline Options and check the boxes to “Display Equation” and “Display R-squared value” on the chart. The result will look like the following figure.




The  $R^2$  value is a measure of the goodness of fit, with a value of 1.0 being the maximum. Even though this trendline has a high  $R^2$  value of 0.956, there is clearly curvature in the data and a line is not the most appropriate function to model the data. Another clue

that the chosen function is not appropriate is that the linear trendline does not go through zero. The Beer-Lambert law predicts an absorbance of zero for a concentration of zero.

If you closed the Format Trendline box, right click on the trendline and choose “Format Trendline...” to reopen it. Select Polynomial in Trendline Options. Changing the order does not make much difference for this data set so we use the simplest function. The following figure shows the fit for a second-order polynomial.



In the following figure I show the deviations between the experimental data and the trendline fit for the second-order polynomial. Comparing the y-scale values to the previous figure shows that the deviations are small. The deviations are random with both positive and negative values and no systematic trend.



2. Determine the Fe concentration in the wine using a linear fit to the standards.

There are multiple ways in Excel to calculate an unknown concentration from a calibration function. The values of the trendline may be typed into cells for the calculation, but they will not update if an error in the calibration data is found and corrected. In the following example, I use the built-in slope and intercept functions of Excel. Note that the spreadsheet formulas have the range of y values first followed by the x values.

For the data in the following cells:

	A	B	C	D	E
15	<b>Table 9.A.1</b>				
16		<b>run</b>	<b>[Fe] (ppm)</b>	<b>A</b>	<b>std dev</b>
17		1	0.00	0.000	0.003
18		2	2.00	0.226	0.001
19		3	4.00	0.404	0.001
20		4	6.00	0.540	0.001
21		5	8.00	0.638	0.002
22		6	10.0	0.700	0.001
23		7	wine	0.148	0.001
24		8	wine + spike	0.305	0.001
25					

Click in a cell below the data table and type:

$$= \text{SLOPE}(D17:D22,C17:C22)$$

Hit Enter and in the next lower cell type:

$$= \text{INTERCEPT}(D17:D22,C17:C22)$$

Hit Enter and in the next lower cell type:

$$= (D23-D30)/D29$$

to calculate the Fe concentration in the wine unknown.

In this last expression, D30 contained the intercept value and D29 contained the slope value. This formula is equivalent to finding the Fe concentration,  $c$ , using:

$$A = 0.148 = (0.0696 \text{ ppm}^{-1})c + 0.0699$$

Add labels and units and the results will look like the following figure:

<b>Unknown using linear calibration function.</b>		
slope =	0.0696	ppm <sup>-1</sup>
intercept =	0.0699	
wine conc =	1.12	ppm Fe

3. Determine the Fe concentration in the wine using a polynomial fit to the standards.

As with a linear function, there are multiple ways in Excel to fit a data set to a polynomial function. The simplest way is to select the desired trendline and retype the function parameters in cells of the worksheet. As with the linear function in the previous example, these values won't update if the data set changes. I will use the Linest function in Excel to fit the calibration data to a second-order polynomial function.

Label three horizontal cells as:

a2                      a1                      a0


Where these are the parameters for a second-order polynomial:

$$y = (a2)x^2 + (a1)x + a0$$

Highlight the three cells below these labels with the mouse and type the following function in the formula bar:

=LINEST(D17:D22,C17:C22^{1,2})

The “^{1,2}” designation instruct the LINEST command to fit the data to a second-order polynomial. You will see something like the following figure as you type. We may omit the last two optional parameters in the function. They are true/false options to set the intercept to zero, [const]=0, and to display statistics such as standard deviations of the fit parameters.<sup>1</sup>



Hold the **Ctrl** and **Shift** keys and then hit **Enter** to create an array of results. The values should be the same as found by the polynomial trendline:

a2	a1	a0
-0.005094866	0.120566518	0.0019375

We now have our calibration function parameters in cells that will update if the calibration data is altered.

<sup>1</sup> To get the standard deviations for the fit parameters, select a 3x5 array of cells, type =LINEST(D17:D22,C17:C22^{1,2},1,1), and press Ctrl-Shift-Enter.

Now that we have the calibration function:

$$y = -0.005095x^2 + 0.1206x + 0.00194$$

or

$$A = -0.005095c^2 + 0.1206c + 0.00194$$

we can enter the absorbance measurements of unknown solutions to determine the analyte concentrations.

Rearranging the expression for the wine measurement gives:

$$A = 0.148 = -0.005095c^2 + 0.1206c + 0.00194$$

$$0 = -0.005095c^2 + 0.1206c - 0.1461$$

The easiest way to solve for just one or two values is to enter the parameters, usually written as  $a$ ,  $b$ , and  $c$ , into an online quadratic calculator. Entering the values from the expression above returns two results, 1.28 and 22.39. Looking at the plot of data, an absorbance of 1.48 corresponds to a concentration near 1 ppm. The 22.39 solution does not make sense, so the correct result is 1.28 ppm Fe.

In the following figure I show how to find an unknown concentration typing in the quadratic equation. Row 38 contains the  $a$ ,  $b$ , and  $c$  parameters, where cell E38 contains the formula:

=E36-D23

	B	C	D	E
32				
33		<b>Unknown using polynomial calibration function.</b>		
34		Linest results:		
35		a2	a1	a0
36		-0.005094866	0.120566518	0.0019375
37		a	b	c
38		-0.0051	0.1206	-0.146
39		wine conc =	1.28	ppm Fe
40				

In cell D39 I entered the following formula to solve the quadratic equation:

$$=(-D38+SQRT(D38^2-(4*C38*E38)))/(2*C38)$$

4. Assuming that the polynomial fit gave the correct answer, determine the error in the result using the linear fit.

Assuming that the true concentration is 1.28 ppm Fe, we can calculate the difference in the linear fit result to get both absolute and relative errors:

$$\text{Error} = 1.28 \text{ ppm} - 1.12 \text{ ppm} = 0.16 \text{ ppm.}$$

$$\text{Relative error} = \frac{0.16 \text{ ppm}}{1.28 \text{ ppm}} \times 100\% = 13\%$$

The error in using a linear calibration function for data that is not linear will depend on the value of the unknown measurement. For our example, the linear function crosses the polynomial line at absorbance values near 0.18 and 0.65 (1.5 ppm and 8.5 ppm, respectively). Measurements that are not near these two points will be in error if using a linear function.

In general, choose an appropriate function that models the expected physical phenomenon. A systematic trend in the plot of the deviations can indicate an incorrect model. If different functions are not significant, stick with the simplest model. A high value for a goodness-of-fit measure does not always indicate a correct model.

5. Calculate the expected absorbance for the wine + spike.  
Determine if a bias occurs for the Fe concentration due to a matrix effect.

The wine is a true unknown. We can only infer the accuracy of a result by validation experiments of similar known samples. One approach to check the accuracy of a measurement is to spike an unknown solution with an additional amount of the analyte. The measured value should increase by an amount proportional to what was added.

The spiked sample had 3.33 ppm of Fe added before measurement. For our unknown result of 1.28 ppm, we expect the spiked wine to contain 4.61 ppm Fe. Calculating the concentration as we did before with the polynomial calibration function gives a concentration of 0.416 ppm iron:

<b>Wine + spike absorbance.</b>		
1.28 ppm + 3.33 ppm =		
	4.61	ppm Fe
<b>Bias in Fe using flame AA.</b>		
a	b	c
-0.0051	0.1206	-0.393
wine+spike conc =	3.90	ppm Fe


This result is more than 10% lower than our expected value. It turns out that there is a known bias for flame AA measurements of iron in an air-acetylene flame. Procedures from instrument manufacturers recommend using the higher temperature nitrous oxide-acetylene flame for iron analysis to reduce interferences.

An additional data set of flame AAS measurements for Mn is in a separate worksheet. The general procedure for analyzing that data is the same as given above. Mn is not prone to interferences and the spiked sample increases by the expected amount.



## 10.B AES

This worksheet contains several tables of data from ICP-AES measurements of metals in aqueous solution. The figure below shows the type of spectral data obtained from the instrument. The software integrates the spectral peaks as shown by the gray bars under the peaks. Each emission line for a given element has an intrinsic “strength” or brightness. If an analyte concentration is high and beyond the linear range of a measurement, using a weaker or less sensitive emission line will provide a more linear measurement range. For the 2.0 ppm solution used to generate the figure, the Ba line at 455.437 nm exceeds the detector range and no peak is shown. For this analyte, at this concentration, we would take measurements from the much weaker line at 233.539 nm.



Concepts	Concept Synopsis
Simultaneous analysis	multi-line emission
sensitivity	Different lines have different sensitivities and can be useful in different analyte concentration ranges.
LOD	Using repetitive measurements to get a measure of noise to calculate limit of detection (LOD).


1. Plot the standard data in Table 10.B.1.

Table 10.B.1 is a selection of data taken from a data file exported from an ICP-AES instrument. Plotting this type of data is not different from what we have already done, but there can be a lot more data to evaluate and analyze.


The quickest way to plot all data is to highlight cells D16:I22 with the mouse and insert a scatter chart on the **Insert** ribbon.

	B	C	D	E	F	G	H	I
14		Table 9.B.1 ICP Atomic Emission of Standard Solutions						
15		Standard Concentration (ppm)						
16		Wavelength	Metal	0.00	0.20	0.40	1.00	2.00
17		267.716	Cr	20.44	6854	14840	36670	73900
18		220.353	Pb	-8.4	348.7	1327	3283	6545
19		213.857	Zn	111.5	8221	15900	40370	81720
20		324.752	Cu	1665.64	41930	82550	202400	402400
21		259.939	Fe	180.88	9480	36580	90790	182700
22		257.610	Mn	-22.66	85370	173000	427400	854900
23								


Doing so produces something like the following figure (in Excel 365):





From **Chart Tools > Design**, click on “Switch Row/Column” to produce six data plots with a legend labeling each metal. You can also access the Switch Row/Column by right clicking in the chart and selecting “Select Data...” Once the chart is generated, you can add trendlines by right clicking on each data set and choosing “Add Trendline...”



Now the chart looks like:




I find the default styles in Excel less than ideal, and I usually do quite a bit of reformatting. You want to produce data plots that show the quality of the data. There is no need for empty space, so adjust the axes to spread the data fully. Likewise, if points are not visible due to overlap, offset data sets or make multiple plots. In this case the different emission lines have different sensitivities, so I will split the data between two charts to group the stronger lines and the weaker lines separately (see following figures). For each data plot I add a trendline and set the intercept to go through zero. Although many of the blank measurements are not exactly zero, setting the intercept to go through the origin makes very little difference in finding the slope of the trendline.





## 2. Plot the Ba standard data in Table 10.B.2.

Plotting two sets of data of very different magnitude is best done in separate charts. I combine them in one chart in the following figure to illustrate the problem. I can't see either data set very well, but the  $R^2$  value does indicate that a linear function fits each of them well. One of the things that I needed to change from the default display was to right-click on the trendline equation for the red data set and change the "Format Trendline Label..." and change the number category to scientific so that I could see more significant figures in the equation.



The following two figures show the data plotted separately. Note that the negative blank measurement does not display in the 455.40 nm data set when I set the minimum y value to zero. However, the trendline does include this point in the fit.



3. Find the average and standard deviation of the replicate measurements in Table 10.B.3. Use these results to determine the LOD for each metal.

There are numerous acceptable methods to determine the limit of detection (LOD) for a given analyte and instrumental method. The simplest method is to determine the signal that is three times the noise above the baseline and convert that signal level to concentration. Random noise will have a Gaussian distribution and we can use the standard deviation,  $\sigma$ , as a measure of the noise. The probability that a random fluctuation occurs  $3\sigma$  from the mean is approximately 1 in 1000. A measured signal that is  $3\sigma$  from the baseline is taken to be due to analyte.

The replicate experiments are made to determine the random variation or noise in these measurements. Calculate the average and standard deviation for the data in each column of Table 10.B.3 by entering the following functions in cells D43 and D44:

= AVERAGE(D36:D41)

= STDEV(D36:D41)

Highlight cells D43 and D44 with the mouse and press Ctrl-C on the keyboard to copy the formulas. Highlight cells E43 through G44 with the mouse and press Ctrl-V to paste these formulas for columns E, F, and G. The results in are in the final figure below.

The LOD will be three times the noise divided by the slope. I find it easier to write the formula for slope to the right of Tables 10.B.1 and 10.B.2. For the Pb data, type in cell J18:

= SLOPE(E18:I18,\$E16:\$I16)

By using an absolute reference, \$E, for the x values, I can copy this cell and paste it to the other rows. Now for the slope of the Pb data I enter "=J18" in cell D45. I do the same for the other elements, changing the cell reference appropriately. These slope values are slightly different from the trendlines, but once I truncate significant figures it will not make any difference.

Now write formulas for LOD in row 46. For the Pb case, I place in cell D46:

=3\*D44/D45.

Copy the formula in cell D46 to cells E46 through G46. For better visibility, I convert from ppm to ppb by multiplying by 1000 in row 47. The next figure shows all of the results:

	B	C	D	E	F	G
33						
34		<b>Table 9.B.3 ICP Atomic Emission of 0.100 ppm Validation Standard</b>				
35		Run #	Pb (220.35)	Zn (213.86)	Ba (233.52)	Ba (455.40)
36		1	333	4064	3223	476122
37		2	333	4128	3311	463503
38		3	311	3999	3094	467987
39		4	377	4027	3293	475400
40		5	355	4188	3104	455123
41		6	366	3822	3299	476104
42						
43		average	346	4038	3221	469040
44		std deviation	25	126	99	8555
45		slope	3335	40821	28138	4790701
46		LOD (ppm)	0.0221	0.0093	0.0106	0.0054
47		LOD (ppb)	22	9	11	5

## 10.C ICP-MS

The process of using ICP-MS data for quantitative elemental analysis is conceptually the same as making quantitative measurements with AAS or ICP-AES. What is different is that ICP-MS instruments can obtain a lot more information. In addition to isotope analysis, coupling a laser or other microprobe sampling method to an ICP-MS allows rapid data collection as a function of spatial dimensions. One such method is known as laser ablation inductively coupled plasma mass spectrometry (LA-ICP-MS). This instrument can make measurements of most of the elements in the periodic table of a solid sample with minimal sample preparation.

This worksheet contains a set of data obtained from a geologic sample containing a melt inclusion. You will plot the data as a function spatial position and calculate average elemental composition inside and outside of the inclusion.


Concepts	Concept Synopsis
Data visualization	Plotting elemental analysis results as a function of spatial dimensions.
Semi-quantitative analysis	A semi-quantitative analysis calibrates the instrument for some elements and interpolates the sensitivity for other similar elements.

This exercise requires some background information on the process of obtaining elemental analytical results using LA-ICP-MS. A solid surface is polished and viewed through a microscope. A spot is selected for analysis and a pulsed laser is turned on for some amount of time. The energy of the laser is sufficient to ablate solid material into the gas phase, which is transported to the ICP plasma by a gas flow. The following figure shows shows LA-ICP-MS data for the circled spot in the image.<sup>2</sup> The y scale is count rate at the detector and the x scale is time. The laser is turned on at approximately 45 s and turned off at 65 s. The measurement is taken as the average of the count rate during the time that the laser is on.

---

<sup>2</sup> Todor Todorov, Erin Marsh, Alan Koenig, *Instrumentation to Characterize Single Inclusions, Laser Ablation-Inductively Coupled Plasma-Mass Spectrometer*, [https://minerals.cr.usgs.gov/dial/LA\\_ICPMS.html](https://minerals.cr.usgs.gov/dial/LA_ICPMS.html).





*Example of LA-ICP-MS ablation spectra from the ablation of a glassy melt inclusion in quartz.*

Excel is useful for making 2-D scatter plots and performing calculations, but it is less useful for manipulating 3-D data. In practice, scientists will use specialized programs to visualize large or complicated data sets. This exercise illustrates some of the capabilities of Excel for these types of surface plots.

1. Create surface plots for the Al and Fe data that is in the 10.C-data worksheet.

In cell C15, type “5” and in cell C16 type “15”. Highlight cells C15:C16 with the mouse. Grab the small square handle at the lower right corner of the highlighted area and drag it down to cell C22. Release and Excel should have filled in values from 5 to 75. Enter “5” and “15” in cells D14 and E14, respectively, highlight the cells, grab the x handle, and drag sideways to cell K14. You should now have the x and y axes for the surface plot. Since the analysis area is 80  $\mu\text{m}$  by 80  $\mu\text{m}$ , I use an 8 $\times$ 8 matrix with a spacing of 10  $\mu\text{m}$ . A finer spacing will show more detail but create a larger data set.

Click on cell D15, type “=”, but don’t hit Enter. Click on the 10.C-data tab, click on cell B7, and now hit return. The following formula:

=’10.C-data!’B7

will appear in cell D15. Edit it to make the column reference absolute:

=’10.C-data!’\$B7

Copy cell D15 to E15 and edit it to add eight to the cell location:

=’10.C-data!’\$B15



Continue copying and editing to fill out row 15. Highlight the row of data and double click on the small square handle at the lower right of the highlighted cells:

	C	D	E	F	G	H	I	J	K
12									
13		Al							
14		5	15	25	35	45	55	65	75
15	5	6764.8	7024.2	7442.0	7134.7	7993.1	8272.6	8220.7	7911.5
16	15								
17	25								

Excel will fill the rest of the rows with data. This process has taken the linear data set and transformed it to a eight-by-eight matrix that Excel can plot as a 3-D surface plot.

Highlight cells C14 to K22 with the mouse. Click on the Insert ribbon and find the surface plot in the Charts group. Hover the mouse over the choices and pick one to create a plot. Click on one of the Surface icons. The figures below show a 3-D plot and a contour plot of the data set. The surface charts cannot be customized as much as the scatter charts. Note that the axes do not extend from 0 to 80, so there is some distortion in this data presentation. You can right-click on the chart and select “3-D Rotation...” to alter


the look of the plot. Move the chart below the data set and edit it as you like. We will copy it to reuse.



To plot another data set, select cells D14:K22 with the mouse and press Ctrl-c on the keyboard to copy the formulas to the clipboard. Move the mouse to M14 and press Ctrl-v on the keyboard to copy the formulas to a new location. Use the mouse to select the data from M15:T22. On the Home ribbon, click on “Find & Select” and choose “Replace...” In the “Find what:” box, type **\$b**. In the “Replace with:” box, type **\$g**. Click on the “Replace All” button. A pop-up box should tell you that Excel made 64 replacements.


Click on the “OK” button to close the replacement dialog box and click on the “Close” button to close the “Find & Replace” dialog. You have now replaced the Al data with data values for Fe. Add labels above the data sets to keep them straight.

Click on the chart that you created for the Al data and press Ctrl-c on the keyboard. Click on cell M24 (anywhere is okay) and press Ctrl-v to make a copy of the chart.



When the chart is selected, the data is highlighted with colored borders. Hover the mouse over the top or bottom borders so the cursor turns into the four-headed move symbol. Click on the border and drag the borders to cover the data in cells M14:T22. The new plot should change to reflect the new data values (see figure on next page).

To quickly visualize the data for the other elements, repeat the replace procedure using “Find & Select > Replace...” In the “Find what:” box, type \$g and in the “Replace with:” box, type \$ with a new column letter. You can see that Excel is not the most convenient program for visualizing 3-D scientific data. Scientists who work with this type of data have specialized programs to take the data files and visualize the data in different ways.




2. Use the semi-quantitative sensitivities in the 10.C-data worksheet to calculate the fractional amounts of Al and Fe both within and outside of the melt inclusion.

Let me provide some background before setting up the Excel calculation. A semi-quantitative analysis in ICP-MS uses calibration functions for each element that are predicted from the measurement of standards of a subset of the elements. The ICP-MS sensitivity increases with atomic mass for similar elements. By similar, I mean that metals will follow an increasing trend with atomic mass, but the halogens and inert gases will deviate significantly from this trend.

This semi-quantitative approach is useful when analyzing a large number of elements even if standards are not available for each individual analyte. Semi-quantitative analysis also avoids the time-consuming task of performing method validation of each individual analyte. The downside is that the measurement results will have a much larger error bar due to the inferred calibration in this approach.

The following figure shows an example of the trend in sensitivity that was determined from standards of selected metals. The sensitivity for other metals can be calculated using the function shown in the trendline equation. Although greater uncertainty will be expected in semi-quantitative measurements, the results can be accurate to the order-of-magnitude level and are often much better.



For convenience, you can click on the two charts and drag them down 9 or 10 rows to make room for the calculation. In the cells below the data set array, type descriptive labels as shown in the following figure:

	C	D	E	F	G	H
23						
24		<b>Aluminum</b>				
25			<b>matrix</b>		<b>inclusion</b>	
26		<b>signal:</b>				
27		<b>sensitivity:</b>				
28		<b>conc:</b>				
29		<b>conversion:</b>				
30		<b>conc:</b>				
31						

Look at the surface plots to choose data points that represent the matrix (the region outside of the inclusion). The signal levels at the front left region of the surface plot are very similar. These points correspond to  $x = 5, 15$  and  $y = 5, 15$ , and we will average these four points as the matrix measurement. In cells E26 through E30, type:

```
=AVERAGE(D15:E16)
='10.C-data'!$B$5
=E26/E27
0.0001
=E28*E29
```

The sensitivity in cell E27 in cps/ppm is in row 5 of the “10.C-data” worksheet. The conversion factor in cell E29 will convert concentration in ppm to concentration in atomic%. Geoscientists will express elemental composition as percent  $\text{Al}_2\text{O}_3$ , NaO,  $\text{SiO}_2$ , etc, but we will not make this additional conversion. Lastly, type appropriate units to the right of the values.

Repeat this calculation for the interior of the inclusion. Type a label such as “inclusion” in cell G25. Looking at the data values, the signal levels around  $x = y = 45$  are fairly uniform. Average these values to represent the center of the inclusion. In cells G26 through G30, type:

```
=AVERAGE(G19:I20)
='10.C-data'!$B$5
=G26/G27
0.0001
=G28*G29
```

	C	D	E	F	G	H
23						
24		<b>Aluminum</b>				
25			<b>matrix</b>		<b>inclusion</b>	
26		<b>signal:</b>	6901 cps		91732 cps	
27		<b>sensitivity:</b>	0.644 cps/ppm		0.644 cps/ppm	
28		<b>conc:</b>	1.07E+04 ppm		1.42E+05 ppm	
29		<b>conversion:</b>	1.0E-04 %/ppm		1.0E-04 %/ppm	
30		<b>conc:</b>	1.1 %		14.2 %	
31						

Make a copy of this table and paste it below the Fe data. Rename the title “Iron” and edit the formulas to perform the same calculation but for the Fe data. For consistency, I average the same spatial data points for the Fe as I selected for the Al. When doing so, it is worthwhile to inspect the data to check for outliers. In this case the selected points appear similar in magnitude. The results are as follows:

	L	M	N	O	P	Q
23						
24		<b>Iron</b>				
25			<b>matrix</b>		<b>inclusion</b>	
26		<b>signal:</b>	110335 cps		41459 cps	
27		<b>sensitivity:</b>	1.490 cps/ppm		1.490 cps/ppm	
28		<b>conc:</b>	7.40E+04 ppm		2.78E+04 ppm	
29		<b>conversion:</b>	1.0E-04 %/ppm		1.0E-04 %/ppm	
30		<b>conc:</b>	7.4 %		2.8 %	
31						

If we wanted to do this calculation for all elements, it would be easier to do on the 10.C data worksheet. The surface plots though help us to choose the data to average to obtain signal levels for the two spatial regions.